

## モータ特性曲線の計算プログラム

### 1 目標応答に対するモータ特性曲線の計算

台車の目標応答（最高でこの程度台車が運動してくればよいという応答）を次の2次遅れ系のステップ応答とした。

$$G(s) = \frac{1}{(1+Ts)^2}, \quad T = 0.1 \text{ [s]}$$

ただし、ステップ入力の大きさは、 $a_{ref} = 0.2 \text{ [m]}$  とする。

この応答に対する台車の速度  $v \text{ [m/s]}$ 、加速度  $a \text{ [m/s}^2\text{]}$ 、モータの回転数  $N \text{ [rpm]}$ 、出力  $P_O \text{ [W]}$ 、トルク  $\tau \text{ [mN}\cdot\text{m]}$  を計算する Scilab プログラムを以下に示す。なお、プログラムの実行には、Scilab ver.4.1.2 を使用した。

```

step-TG70A.sce

// step_TG70A.sce 2009/07/23
// 設計の目標となるステップ応答に対するモータの諸特性の計算

s=%s;
T=0.1; // 閉ループ系の時定数 [s]
M=1; // 駆動系の等価質量 [kg]
r=0.0075; // ピニオンギアの半径 [m]
Gear_ratio=4; // ギア比の逆数
a_ref=0.2; // ステップ入力の大きさ [m]

Gv=s/(1+T*s)^2; // 目標入力から台車速度までの伝達関数
Ga=s*Gv; // 目標入力から台車加速度までの伝達関数
t=0:0.001:1; // 応答の時間軸
Gvs=syslin('c',Gv); // Gv を線形連続時間系として定義
Gas=syslin('c',Ga); // Ga "
v=a_ref*csim('step',t,Gvs); // 台車速度のステップ応答
a=a_ref*csim('step',t,Gas); // 台車加速度のステップ応答

N=v/(2*pi*r)*60*Gear_ratio; // モータの回転数 [rpm]
Po=M*a.*v; // " 出力 [W]=[Nm/s]
tau=M*a*r/Gear_ratio*1000; // " トルク [mN*m]

xset('window',0) // 速度と加速度のグラフ

```

```

plot2d(t,v,2)
plot2d(t,a,5)
xgrid(4)

xset('window',1)           // 回転数のグラフ
plot2d(t,N,5)
xgrid(4)

xset('window',2)           // 出力のグラフ
plot2d(t,Po,5)
xgrid(4)

xset('window',3)           // トルクのグラフ
plot2d(t,tau,5)
xgrid(4)

```

上記のプログラムにおいて、 $M$  は駆動系の等価質量である。モータ、ギア、台車等を考慮して、実際の台車質量の 2 倍以上の質量とした。

モータの最大回転数が使用するモータの定格回転数を超えないように、また、出力とトルクもモータの負荷特性図から読み取れる値を超えないようにモータとギア比を選定し、ピニオンギアの半径を設計する。

本台車系実験装置の場合、最大回転数はモータの定格回転数の約 75 % である。逆に、これが少なすぎると、モータの能力を十分に利用していないことになる。

図 1, 2, 3 にそれぞれモータの回転数、出力、トルクのグラフを示す。

圧力角の関係で、負荷側からのトルクがモータの回転軸に伝わらない構造のギアドモータについては、上記の計算式および台車制御サンプルプログラムの制御則が使えないので注意する。




図 1: 目標応答に対するモータの回転数




図 2: 目標応答に対するモータの出力




図 3: 目標応答に対するモータのトルク